# $Id: PKGBUILD 182923 2013-04-16 07:16:05Z tpowa $
# Maintainer: Tobias Powalowski <tpowa@archlinux.org>

pkgname=lensfun
pkgver=0.2.7
pkgrel=1
pkgdesc="Database of photographic lenses and a library that allows advanced access to the database"
arch=(i686 x86_64)
url="http://lensfun.berlios.de/"
license=('LGPL3')
depends=('glibc' 'glib2')
makedepends=('python2' 'libpng' 'doxygen')
source=("http://download.berlios.de/${pkgname}/${pkgname}-${pkgver}.tar.bz2")

prepare() {
    cd "${srcdir}/${pkgname}-${pkgver}"
    sed -i 's|#!/usr/bin/python|#!/usr/bin/python2|' configure
}

build() {
    cd "${srcdir}/${pkgname}-${pkgver}"
   ./configure \
        --prefix=/usr \
        --libdir=/usr/lib
    make all
}

package() {
    cd "${srcdir}/${pkgname}-${pkgver}"
    make INSTALL_PREFIX="$pkgdir" install
}
md5sums=('6061cc2d00a42d3e62512878426fc841')
